Cálculo 3 - 2024.2

P2 (segunda prova)

Eduardo Ochs - RCN/PURO/UFF
http://anggtwu.net/2024.2-C3.html

Links

http://anggtwu.net/e/maxima.e.html#2024.2-C3-P2
(find-es "maxima" "2024.2-C3-P2")

Questão 1

(Total: 10.0 pts)

Sejam:

$$\begin{array}{rcl} F(x,y) &=& x+y^2 \\ G(x,y) &=& 2x+y \\ r &=& G^{-1}(-2) \\ A &=& \{-2,-1,0,1,2\}^2 \\ B &=& [-2,2]^2 \\ C &=& \{\,(x,y)\in B \mid G(x,y)\geq -2\,\} \\ D &=& \{\,(x,y)\in B \mid G(x,y) = -2\,\} \end{array}$$

$$\begin{array}{rcccl} F_B &:& B &\to& \mathbb{R} \\ && (x,y) &\mapsto& F(x,y) \\ F_C &:& C &\to& \mathbb{R} \\ && (x,y) &\mapsto& F(x,y) \end{array}$$

a) **(0.5 pts)** Faça o diagrama de numerozinhos da função F. Desenhe um numerozinho para cada ponto de A.

b) **(0.5 pts)** Idem, mas para a função G.

c) **(0.5 pts)** Desenhe as curvas de nível da G para $z=-2,-1,0,1,2$.

d) **(0.5 pts)** Desenhe os conjuntos r,A,B,C,D.

e) **(0.5 pts)** Desenhe as curvas de nível da F para $z=-1,0,1,2$.

f) **(0.5 pts)** Desenhe as curvas de nível da F_B para $z=-1,0,1,2$. Inclua a fronteira do domínio da F_B no seu desenho pra leitor conseguir ver até onde essas curvas de nível vão.

g) **(1.0 pts)** Desenhe as curvas de nível da F_C para $z=-1,0,1,2$. Inclua a fronteira do domínio da F_C no seu desenho pra leitor conseguir ver até onde essas curvas de nível vão.

h) **(1.0 pts)** Dê coordenadas *aproximadas* para os máximos e mínimos locais da função F_C.

Nos próximos itens você vai tentar fazer um truque que o Bortolossi ensina no capítulo 12 dele, em que ele mostra como encontrar máximos e mínimos locais na fronteira a partir de pontos na fronteira em que dois vetores gradientes são paralelos – e com isso você vai conseguir fazer uma versão melhorada do seu item (h).

Digamos que:

$$\begin{array}{rcl} d(x) &=& ax+b \\ s &=& \{\,(x,y)\in\mathbb{R}^2 \mid y=d(x)\,\} \\ P(x) &=& (x,d(x)) \end{array}$$

tais que $s=r$.

Isso (↑) é uma *definição indireta*. O autor – no caso, eu – não está dizendo quais são os valores de a e b, e o leitor – no caso, você – vai ter que se virar pra descobrir.

i) **(0.5 pts)** Encontre a e b por chutar e testar. Você vai começar fazendo vários chutes-e-testes assim:

Se $(a,b)=(0,1)$ então $s=$  e $s\neq r$ =(

Se $(a,b)=(1,1)$ então $s=$  e $s\neq r$ =(

j) **(0.5 pts)** Calcule $\nabla F(x,y)$, $\nabla G(x,y)$, $\nabla F(P(x))$ e $\nabla G(P(x))$.

k) **(1.0 pts)** Faça uma cópia do seu desenho do item (g) incluindo os vetores $P(x)+\nabla F(P(x))$ e $P(x)+\nabla G(P(x))$ para os seguintes valores de x: $x=-1.5$, $x=-1$, $x=-0.5$.

l) **(1.0 pts)** Encontre o valor de x que faz com que $\nabla F(P(x))$ e $\nabla G(P(x))$ fiquem paralelos. Sugestão: faça uma tabela e encontre ele por chutar e testar. Chame esse valor de x de x_m.

m) **(1.0 pts)** Faça uma cópia do seu desenho do item (g) incluindo os vetores $P(x)+\nabla F(P(x))$ e $P(x)+\nabla G(P(x))$ para $x=x_m$. Faça ele BEM grande e capriche!

n) **(1.0 pts)** Calcule $F(D)$ e $F(C)$..

Lembre que se $F:\mathbb{R}^2 \to \mathbb{R}$, $A \subseteq \mathbb{R}^2$ e $B \subseteq \mathbb{R}$ então:

$$\begin{array}{rcl} F(A) &=& \{\, F(x,y) \mid (x,y) \in A \,\} \\ F^{-1}(b) &=& \{\, (x,y) \in \mathbb{R}^2 \mid F(x,y) = b \,\} \\ F^{-1}(B) &=& \{\, (x,y) \in \mathbb{R}^2 \mid F(x,y) \in B \,\} \end{array}$$

e que quando a gente escreve

$$\begin{array}{rcccl} G &:& A & \to & \mathbb{R} \\ && (x,y) & \mapsto & F(x,y) \end{array}$$

isso define uma função G que se comporta como a F dentro do conjunto A, mas o domínio dessa G é só o conjunto A – quando a G recebe um ponto de $\mathbb{R}^2 \backslash A$ ela dá um erro.

Lembre que dois vetores $\vec{v}$ e $\vec{w}$ não nulos são *paralelos* quando $\exists \lambda \in \mathbb{R}. \vec{v} = \lambda \vec{w}$. Por exemplo, $\overrightarrow{(2,3)}$ e $\overrightarrow{(20,30)}$ são paralelos mas $\overrightarrow{(2,3)}$ e $\overrightarrow{(5,4)}$ não são paralelos.

```
(%i1) [xmin,xmax, ymin,ymax] : [-2,2, -2,2]$
(%i2) numsB(expr) :=
        apply(matrix,
              makelist(makelist(ev(expr), x,-2,2),
                       y, seqby(2,-2,-1)))$

(%i3) F : x + y^2$
(%i4) G : 2*x + y$
(%i5) define(F(x,y), F);
```

(%o5)

$$F(x,y) := y^2 + x$$

```
(%i6) define(G(x,y), G);
```

(%o6)

$$G(x,y) := y + 2\,x$$

```
(%i7)
      /* (a): 0.5 pts */
      numsB(F);
```

(%o7)

$$\begin{pmatrix} 2 & 3 & 4 & 5 & 6 \\ -1 & 0 & 1 & 2 & 3 \\ -2 & -1 & 0 & 1 & 2 \\ -1 & 0 & 1 & 2 & 3 \\ 2 & 3 & 4 & 5 & 6 \end{pmatrix}$$

```
(%i8)
      /* (b): 0.5 pts */
      numsB(G);
```

(%o8)

$$\begin{pmatrix} -2 & 0 & 2 & 4 & 6 \\ -3 & -1 & 1 & 3 & 5 \\ -4 & -2 & 0 & 2 & 4 \\ -5 & -3 & -1 & 1 & 3 \\ -6 & -4 & -2 & 0 & 2 \end{pmatrix}$$

```
(%i9) /* (c): 0.5 pts */
      G_B : [myimp1(G=-2, lc(blue)),
             myimp1(G=-1, lc(forest_green)),
             myimp1(G=0,  lc(gold)),
             myimp1(G=1,  lc(orange)),
             myimp1(G=2,  lc(red))]$
```

```
(%i10) myqdrawp(xyrange(), G_B);

       /* (d): 0.5 pts */
```

(%o10)




```
(%i11) D_B : [myimp1(G=-2, lc(gray))]$

(%i12) myqdrawp(xyrange(), D_B);
```

(%o12)




```
(%i13)
       /* (e): 0.5 pts */
       /* (f): 0.5 pts */
       F_B : [myimp1(F=-1, lc(blue)),
              myimp1(F=0,  lc(forest_green)),
              myimp1(F=1,  lc(orange)),
              myimp1(F=2,  lc(red))]$
(%i14) myqdrawp(xyrange(), F_B);
```

(%o14)



```
(%i15) /* (g): 1.0 pts */
       F_ys(z) := block([ys],
         ys : solve([G=-2,F=z],[x,y]),
         [rhs(ys[1][2]), rhs(ys[2][2])])$
(%i16) F_miny(z) := apply('min, F_ys(z))$
(%i17) F_maxy(z) := apply('max, F_ys(z))$
(%i18) F_C_z(z,color) := myapply_fl('imp1,
              F=z, x,-2,2, y,F_miny(z),F_maxy(z), lc(color))$
(%i19) F_C : [F_C_z(-1, blue),
              F_C_z( 0, forest_green),
              F_C_z( 1, orange),
              F_C_z( 2, red)]$
(%i20) myqdrawp(xyrange(), D_B, F_C);
```

(%o20)




```
(%i21) /* (h): 1.0 pts */
       [F(-1,0), F(2,-2), F(2,2)];
```

(%o21)

$$[-1, 6, 6]$$

```
(%i22) /* (i): 0.5 pts */
                          G=-2;
```

(%o22)

$$y + 2\,x = -2$$

```
(%i23)                solve(G=-2, y);
```

(%o23)

$$[y = -(2\,x) - 2]$$

```
(%i24)                solve(G=-2, y)[1];
```

(%o24)

$$y = -(2\,x) - 2$$

```
(%i25)             rhs(solve(G=-2, y)[1]);
```

(%o25)

$$-(2\,x) - 2$$

```
(%i26) define(d(x), rhs(solve(G=-2, y)[1]));
```

(%o26)

$$d\,(x) := -(2\,x) - 2$$

```
(%i27) define(P(x), [x,d(x)]);
```

(%o27)

$$P\,(x) := [x, -(2\,x) - 2]$$

```
(%i28) [a,b] : [-2,-2];
```

(%o28)

$$[-2, -2]$$

```
(%i29) /* (j): 0.5 pts */
       grad(F) := [diff(F,x), diff(F,y)]$
(%i30) define(gradF (x,y), grad(F));
```
(%o30)

$$\mathrm{gradF}\,(x,y) := [1, 2\,y]$$

```
(%i31) define(gradG (x,y), grad(G));
```
(%o31)

$$\mathrm{gradG}\,(x,y) := [2, 1]$$

```
(%i32) define(gradFP(x), apply('gradF, P(x)));
```
(%o32)

$$\mathrm{gradFP}\,(x) := [1, 2\,(-(2\,x) - 2)]$$

```
(%i33) define(gradGP(x), apply('gradG, P(x)));
```
(%o33)

$$\mathrm{gradGP}\,(x) := [2, 1]$$

```
(%i34) /* (k): 1.0 pts */
       gradsP(x) := [myPv_c(P(x),gradGP(x)/3, blue),
                     myPv_c(P(x),gradFP(x)/3, dark_violet)]$
(%i35) gradsP_3  : [gradsP(-1.5), gradsP(-1), gradsP(-0.5)]$
(%i36) myqdrawp(xyrange(), D_B, F_C, gradsP_3);
```
(%o36)




```
(%i37) /* (l): 1.0 pts */
       eq1 : 2*gradFP(x)    = gradGP(x);
```
(%o37)

$$[2, 4\,(-(2\,x) - 2)] = [2, 1]$$

```
(%i38) eq2 : 2*gradFP(x)[2] = gradGP(x)[2];
```
(%o38)

$$4\,(-(2\,x) - 2) = 1$$

```
(%i39)         solve(eq2, x);
```
(%o39)

$$\left[x = -\left(\frac{9}{8}\right)\right]$$

```
(%i40)         solve(eq2, x)[1];
```
(%o40)

$$x = -\left(\frac{9}{8}\right)$$

```
(%i41) xm : rhs(solve(eq2, x)[1]);  /*  -9/8  */
```
(%o41)

$$-\left(\frac{9}{8}\right)$$

```
(%i42) ym : d(xm);                  /*   1/4  */
```
(%o42)

$$\frac{1}{4}$$

```
(%i43) Fm : F(xm,ym);               /* -17/16 */
```
(%o43)

$$-\left(\frac{17}{16}\right)$$

```
(%i44) /* (m): 1.0 pts */
(%i44) gradsP_xm : [gradsP(xm)]$
(%i45) myqdrawp(xyrange(), D_B, F_C, gradsP_xm);
(%o45)
```



```
(%i46) /* (n): 1.0 pts */
       Fnw : F(-2,d(-2));
(%o46)
```

$$2$$

```
(%i47) Fsw : F(0,d(0));
(%o47)
```

$$4$$

```
(%i48) Fne : F(2,2);
(%o48)
```

$$6$$

```
(%i49) Fse : F(2,-2);
(%o49)
```

$$6$$

```
(%i50)      [Fm, Fnw, Fsw];
(%o50)
```

$$\left[-\left(\frac{17}{16}\right), 2, 4\right]$$

```
(%i51) FD : [Fm, max(Fnw, Fsw)];
(%o51)
```

$$\left[-\left(\frac{17}{16}\right), 4\right]$$

```
(%i52) FC : [Fm, max(Fnw, Fsw, Fne, Fse)];
(%o52)
```

$$\left[-\left(\frac{17}{16}\right), 6\right]$$